// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/token/ERC20/extensions/draft-IERC20Permit.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/draft-IERC20Permit.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../extensions/draft-IERC20Permit.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address-functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Strings.sol)

pragma solidity ^0.8.0;

import "./math/Math.sol";

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        unchecked {
            uint256 length = Math.log10(value) + 1;
            string memory buffer = new string(length);
            uint256 ptr;
            /// @solidity memory-safe-assembly
            assembly {
                ptr := add(buffer, add(32, length))
            }
            while (true) {
                ptr--;
                /// @solidity memory-safe-assembly
                assembly {
                    mstore8(ptr, byte(mod(value, 10), _SYMBOLS))
                }
                value /= 10;
                if (value == 0) break;
            }
            return buffer;
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        unchecked {
            return toHexString(value, Math.log256(value) + 1);
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }
}

// File: @openzeppelin/contracts/utils/cryptography/ECDSA.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/cryptography/ECDSA.sol)

pragma solidity ^0.8.0;

import "../Strings.sol";

/**
 * @dev Elliptic Curve Digital Signature Algorithm (ECDSA) operations.
 *
 * These functions can be used to verify that a message was signed by the holder
 * of the private keys of a given address.
 */
library ECDSA {
    enum RecoverError {
        NoError,
        InvalidSignature,
        InvalidSignatureLength,
        InvalidSignatureS,
        InvalidSignatureV // Deprecated in v4.8
    }

    function _throwError(RecoverError error) private pure {
        if (error == RecoverError.NoError) {
            return; // no error: do nothing
        } else if (error == RecoverError.InvalidSignature) {
            revert("ECDSA: invalid signature");
        } else if (error == RecoverError.InvalidSignatureLength) {
            revert("ECDSA: invalid signature length");
        } else if (error == RecoverError.InvalidSignatureS) {
            revert("ECDSA: invalid signature 's' value");
        }
    }

    /**
     * @dev Returns the address that signed a hashed message (`hash`) with
     * `signature` or error string. This address can then be used for verification purposes.
     *
     * The `ecrecover` EVM opcode allows for malleable (non-unique) signatures:
     * this function rejects them by requiring the `s` value to be in the lower
     * half order, and the `v` value to be either 27 or 28.
     *
     * IMPORTANT: `hash` _must_ be the result of a hash operation for the
     * verification to be secure: it is possible to craft signatures that
     * recover to arbitrary addresses for non-hashed data. A safe way to ensure
     * this is by receiving a hash of the original message (which may otherwise
     * be too long), and then calling {toEthSignedMessageHash} on it.
     *
     * Documentation for signature generation:
     * - with https://web3js.readthedocs.io/en/v1.3.4/web3-eth-accounts.html#sign[Web3.js]
     * - with https://docs.ethers.io/v5/api/signer/#Signer-signMessage[ethers]
     *
     * _Available since v4.3._
     */
    function tryRecover(bytes32 hash, bytes memory signature) internal pure returns (address, RecoverError) {
        if (signature.length == 65) {
            bytes32 r;
            bytes32 s;
            uint8 v;
            // ecrecover takes the signature parameters, and the only way to get them
            // currently is to use assembly.
            /// @solidity memory-safe-assembly
            assembly {
                r := mload(add(signature, 0x20))
                s := mload(add(signature, 0x40))
                v := byte(0, mload(add(signature, 0x60)))
            }
            return tryRecover(hash, v, r, s);
        } else {
            return (address(0), RecoverError.InvalidSignatureLength);
        }
    }

    /**
     * @dev Returns the address that signed a hashed message (`hash`) with
     * `signature`. This address can then be used for verification purposes.
     *
     * The `ecrecover` EVM opcode allows for malleable (non-unique) signatures:
     * this function rejects them by requiring the `s` value to be in the lower
     * half order, and the `v` value to be either 27 or 28.
     *
     * IMPORTANT: `hash` _must_ be the result of a hash operation for the
     * verification to be secure: it is possible to craft signatures that
     * recover to arbitrary addresses for non-hashed data. A safe way to ensure
     * this is by receiving a hash of the original message (which may otherwise
     * be too long), and then calling {toEthSignedMessageHash} on it.
     */
    function recover(bytes32 hash, bytes memory signature) internal pure returns (address) {
        (address recovered, RecoverError error) = tryRecover(hash, signature);
        _throwError(error);
        return recovered;
    }

    /**
     * @dev Overload of {ECDSA-tryRecover} that receives the `r` and `vs` short-signature fields separately.
     *
     * See https://eips.ethereum.org/EIPS/eip-2098[EIP-2098 short signatures]
     *
     * _Available since v4.3._
     */
    function tryRecover(
        bytes32 hash,
        bytes32 r,
        bytes32 vs
    ) internal pure returns (address, RecoverError) {
        bytes32 s = vs & bytes32(0x7fffffffffffffffffffffffffffffffffffffffffffffffffffffffffffffff);
        uint8 v = uint8((uint256(vs) >> 255) + 27);
        return tryRecover(hash, v, r, s);
    }

    /**
     * @dev Overload of {ECDSA-recover} that receives the `r and `vs` short-signature fields separately.
     *
     * _Available since v4.2._
     */
    function recover(
        bytes32 hash,
        bytes32 r,
        bytes32 vs
    ) internal pure returns (address) {
        (address recovered, RecoverError error) = tryRecover(hash, r, vs);
        _throwError(error);
        return recovered;
    }

    /**
     * @dev Overload of {ECDSA-tryRecover} that receives the `v`,
     * `r` and `s` signature fields separately.
     *
     * _Available since v4.3._
     */
    function tryRecover(
        bytes32 hash,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal pure returns (address, RecoverError) {
        // EIP-2 still allows signature malleability for ecrecover(). Remove this possibility and make the signature
        // unique. Appendix F in the Ethereum Yellow paper (https://ethereum.github.io/yellowpaper/paper.pdf), defines
        // the valid range for s in (301): 0 < s < secp256k1n ÷ 2 + 1, and for v in (302): v ∈ {27, 28}. Most
        // signatures from current libraries generate a unique signature with an s-value in the lower half order.
        //
        // If your library generates malleable signatures, such as s-values in the upper range, calculate a new s-value
        // with 0xFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFEBAAEDCE6AF48A03BBFD25E8CD0364141 - s1 and flip v from 27 to 28 or
        // vice versa. If your library also generates signatures with 0/1 for v instead 27/28, add 27 to v to accept
        // these malleable signatures as well.
        if (uint256(s) > 0x7FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF5D576E7357A4501DDFE92F46681B20A0) {
            return (address(0), RecoverError.InvalidSignatureS);
        }

        // If the signature is valid (and not malleable), return the signer address
        address signer = ecrecover(hash, v, r, s);
        if (signer == address(0)) {
            return (address(0), RecoverError.InvalidSignature);
        }

        return (signer, RecoverError.NoError);
    }

    /**
     * @dev Overload of {ECDSA-recover} that receives the `v`,
     * `r` and `s` signature fields separately.
     */
    function recover(
        bytes32 hash,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal pure returns (address) {
        (address recovered, RecoverError error) = tryRecover(hash, v, r, s);
        _throwError(error);
        return recovered;
    }

    /**
     * @dev Returns an Ethereum Signed Message, created from a `hash`. This
     * produces hash corresponding to the one signed with the
     * https://eth.wiki/json-rpc/API#eth_sign[`eth_sign`]
     * JSON-RPC method as part of EIP-191.
     *
     * See {recover}.
     */
    function toEthSignedMessageHash(bytes32 hash) internal pure returns (bytes32) {
        // 32 is the length in bytes of hash,
        // enforced by the type signature above
        return keccak256(abi.encodePacked("\x19Ethereum Signed Message:\n32", hash));
    }

    /**
     * @dev Returns an Ethereum Signed Message, created from `s`. This
     * produces hash corresponding to the one signed with the
     * https://eth.wiki/json-rpc/API#eth_sign[`eth_sign`]
     * JSON-RPC method as part of EIP-191.
     *
     * See {recover}.
     */
    function toEthSignedMessageHash(bytes memory s) internal pure returns (bytes32) {
        return keccak256(abi.encodePacked("\x19Ethereum Signed Message:\n", Strings.toString(s.length), s));
    }

    /**
     * @dev Returns an Ethereum Signed Typed Data, created from a
     * `domainSeparator` and a `structHash`. This produces hash corresponding
     * to the one signed with the
     * https://eips.ethereum.org/EIPS/eip-712[`eth_signTypedData`]
     * JSON-RPC method as part of EIP-712.
     *
     * See {recover}.
     */
    function toTypedDataHash(bytes32 domainSeparator, bytes32 structHash) internal pure returns (bytes32) {
        return keccak256(abi.encodePacked("\x19\x01", domainSeparator, structHash));
    }
}

// File: @openzeppelin/contracts/utils/cryptography/EIP712.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/cryptography/EIP712.sol)

pragma solidity ^0.8.0;

import "./ECDSA.sol";

/**
 * @dev https://eips.ethereum.org/EIPS/eip-712[EIP 712] is a standard for hashing and signing of typed structured data.
 *
 * The encoding specified in the EIP is very generic, and such a generic implementation in Solidity is not feasible,
 * thus this contract does not implement the encoding itself. Protocols need to implement the type-specific encoding
 * they need in their contracts using a combination of `abi.encode` and `keccak256`.
 *
 * This contract implements the EIP 712 domain separator ({_domainSeparatorV4}) that is used as part of the encoding
 * scheme, and the final step of the encoding to obtain the message digest that is then signed via ECDSA
 * ({_hashTypedDataV4}).
 *
 * The implementation of the domain separator was designed to be as efficient as possible while still properly updating
 * the chain id to protect against replay attacks on an eventual fork of the chain.
 *
 * NOTE: This contract implements the version of the encoding known as "v4", as implemented by the JSON RPC method
 * https://docs.metamask.io/guide/signing-data.html[`eth_signTypedDataV4` in MetaMask].
 *
 * _Available since v3.4._
 */
abstract contract EIP712 {
    /* solhint-disable var-name-mixedcase */
    // Cache the domain separator as an immutable value, but also store the chain id that it corresponds to, in order to
    // invalidate the cached domain separator if the chain id changes.
    bytes32 private immutable _CACHED_DOMAIN_SEPARATOR;
    uint256 private immutable _CACHED_CHAIN_ID;
    address private immutable _CACHED_THIS;

    bytes32 private immutable _HASHED_NAME;
    bytes32 private immutable _HASHED_VERSION;
    bytes32 private immutable _TYPE_HASH;

    /* solhint-enable var-name-mixedcase */

    /**
     * @dev Initializes the domain separator and parameter caches.
     *
     * The meaning of `name` and `version` is specified in
     * https://eips.ethereum.org/EIPS/eip-712#definition-of-domainseparator[EIP 712]:
     *
     * - `name`: the user readable name of the signing domain, i.e. the name of the DApp or the protocol.
     * - `version`: the current major version of the signing domain.
     *
     * NOTE: These parameters cannot be changed except through a xref:learn::upgrading-smart-contracts.adoc[smart
     * contract upgrade].
     */
    constructor(string memory name, string memory version) {
        bytes32 hashedName = keccak256(bytes(name));
        bytes32 hashedVersion = keccak256(bytes(version));
        bytes32 typeHash = keccak256(
            "EIP712Domain(string name,string version,uint256 chainId,address verifyingContract)"
        );
        _HASHED_NAME = hashedName;
        _HASHED_VERSION = hashedVersion;
        _CACHED_CHAIN_ID = block.chainid;
        _CACHED_DOMAIN_SEPARATOR = _buildDomainSeparator(typeHash, hashedName, hashedVersion);
        _CACHED_THIS = address(this);
        _TYPE_HASH = typeHash;
    }

    /**
     * @dev Returns the domain separator for the current chain.
     */
    function _domainSeparatorV4() internal view returns (bytes32) {
        if (address(this) == _CACHED_THIS && block.chainid == _CACHED_CHAIN_ID) {
            return _CACHED_DOMAIN_SEPARATOR;
        } else {
            return _buildDomainSeparator(_TYPE_HASH, _HASHED_NAME, _HASHED_VERSION);
        }
    }

    function _buildDomainSeparator(
        bytes32 typeHash,
        bytes32 nameHash,
        bytes32 versionHash
    ) private view returns (bytes32) {
        return keccak256(abi.encode(typeHash, nameHash, versionHash, block.chainid, address(this)));
    }

    /**
     * @dev Given an already https://eips.ethereum.org/EIPS/eip-712#definition-of-hashstruct[hashed struct], this
     * function returns the hash of the fully encoded EIP712 message for this domain.
     *
     * This hash can be used together with {ECDSA-recover} to obtain the signer of a message. For example:
     *
     * ```solidity
     * bytes32 digest = _hashTypedDataV4(keccak256(abi.encode(
     *     keccak256("Mail(address to,string contents)"),
     *     mailTo,
     *     keccak256(bytes(mailContents))
     * )));
     * address signer = ECDSA.recover(digest, signature);
     * ```
     */
    function _hashTypedDataV4(bytes32 structHash) internal view virtual returns (bytes32) {
        return ECDSA.toTypedDataHash(_domainSeparatorV4(), structHash);
    }
}

// File: @openzeppelin/contracts/utils/cryptography/draft-EIP712.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/cryptography/draft-EIP712.sol)

pragma solidity ^0.8.0;

// EIP-712 is Final as of 2022-08-11. This file is deprecated.

import "./EIP712.sol";

// File: @openzeppelin/contracts/utils/math/Math.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/math/Math.sol)

pragma solidity ^0.8.0;

/**
 * @dev Standard math utilities missing in the Solidity language.
 */
library Math {
    enum Rounding {
        Down, // Toward negative infinity
        Up, // Toward infinity
        Zero // Toward zero
    }

    /**
     * @dev Returns the largest of two numbers.
     */
    function max(uint256 a, uint256 b) internal pure returns (uint256) {
        return a > b ? a : b;
    }

    /**
     * @dev Returns the smallest of two numbers.
     */
    function min(uint256 a, uint256 b) internal pure returns (uint256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two numbers. The result is rounded towards
     * zero.
     */
    function average(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b) / 2 can overflow.
        return (a & b) + (a ^ b) / 2;
    }

    /**
     * @dev Returns the ceiling of the division of two numbers.
     *
     * This differs from standard division with `/` in that it rounds up instead
     * of rounding down.
     */
    function ceilDiv(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b - 1) / b can overflow on addition, so we distribute.
        return a == 0 ? 0 : (a - 1) / b + 1;
    }

    /**
     * @notice Calculates floor(x * y / denominator) with full precision. Throws if result overflows a uint256 or denominator == 0
     * @dev Original credit to Remco Bloemen under MIT license (https://xn--2-umb.com/21/muldiv)
     * with further edits by Uniswap Labs also under MIT license.
     */
    function mulDiv(
        uint256 x,
        uint256 y,
        uint256 denominator
    ) internal pure returns (uint256 result) {
        unchecked {
            // 512-bit multiply [prod1 prod0] = x * y. Compute the product mod 2^256 and mod 2^256 - 1, then use
            // use the Chinese Remainder Theorem to reconstruct the 512 bit result. The result is stored in two 256
            // variables such that product = prod1 * 2^256 + prod0.
            uint256 prod0; // Least significant 256 bits of the product
            uint256 prod1; // Most significant 256 bits of the product
            assembly {
                let mm := mulmod(x, y, not(0))
                prod0 := mul(x, y)
                prod1 := sub(sub(mm, prod0), lt(mm, prod0))
            }

            // Handle non-overflow cases, 256 by 256 division.
            if (prod1 == 0) {
                return prod0 / denominator;
            }

            // Make sure the result is less than 2^256. Also prevents denominator == 0.
            require(denominator > prod1);

            ///////////////////////////////////////////////
            // 512 by 256 division.
            ///////////////////////////////////////////////

            // Make division exact by subtracting the remainder from [prod1 prod0].
            uint256 remainder;
            assembly {
                // Compute remainder using mulmod.
                remainder := mulmod(x, y, denominator)

                // Subtract 256 bit number from 512 bit number.
                prod1 := sub(prod1, gt(remainder, prod0))
                prod0 := sub(prod0, remainder)
            }

            // Factor powers of two out of denominator and compute largest power of two divisor of denominator. Always >= 1.
            // See https://cs.stackexchange.com/q/138556/92363.

            // Does not overflow because the denominator cannot be zero at this stage in the function.
            uint256 twos = denominator & (~denominator + 1);
            assembly {
                // Divide denominator by twos.
                denominator := div(denominator, twos)

                // Divide [prod1 prod0] by twos.
                prod0 := div(prod0, twos)

                // Flip twos such that it is 2^256 / twos. If twos is zero, then it becomes one.
                twos := add(div(sub(0, twos), twos), 1)
            }

            // Shift in bits from prod1 into prod0.
            prod0 |= prod1 * twos;

            // Invert denominator mod 2^256. Now that denominator is an odd number, it has an inverse modulo 2^256 such
            // that denominator * inv = 1 mod 2^256. Compute the inverse by starting with a seed that is correct for
            // four bits. That is, denominator * inv = 1 mod 2^4.
            uint256 inverse = (3 * denominator) ^ 2;

            // Use the Newton-Raphson iteration to improve the precision. Thanks to Hensel's lifting lemma, this also works
            // in modular arithmetic, doubling the correct bits in each step.
            inverse *= 2 - denominator * inverse; // inverse mod 2^8
            inverse *= 2 - denominator * inverse; // inverse mod 2^16
            inverse *= 2 - denominator * inverse; // inverse mod 2^32
            inverse *= 2 - denominator * inverse; // inverse mod 2^64
            inverse *= 2 - denominator * inverse; // inverse mod 2^128
            inverse *= 2 - denominator * inverse; // inverse mod 2^256

            // Because the division is now exact we can divide by multiplying with the modular inverse of denominator.
            // This will give us the correct result modulo 2^256. Since the preconditions guarantee that the outcome is
            // less than 2^256, this is the final result. We don't need to compute the high bits of the result and prod1
            // is no longer required.
            result = prod0 * inverse;
            return result;
        }
    }

    /**
     * @notice Calculates x * y / denominator with full precision, following the selected rounding direction.
     */
    function mulDiv(
        uint256 x,
        uint256 y,
        uint256 denominator,
        Rounding rounding
    ) internal pure returns (uint256) {
        uint256 result = mulDiv(x, y, denominator);
        if (rounding == Rounding.Up && mulmod(x, y, denominator) > 0) {
            result += 1;
        }
        return result;
    }

    /**
     * @dev Returns the square root of a number. If the number is not a perfect square, the value is rounded down.
     *
     * Inspired by Henry S. Warren, Jr.'s "Hacker's Delight" (Chapter 11).
     */
    function sqrt(uint256 a) internal pure returns (uint256) {
        if (a == 0) {
            return 0;
        }

        // For our first guess, we get the biggest power of 2 which is smaller than the square root of the target.
        //
        // We know that the "msb" (most significant bit) of our target number `a` is a power of 2 such that we have
        // `msb(a) <= a < 2*msb(a)`. This value can be written `msb(a)=2**k` with `k=log2(a)`.
        //
        // This can be rewritten `2**log2(a) <= a < 2**(log2(a) + 1)`
        // → `sqrt(2**k) <= sqrt(a) < sqrt(2**(k+1))`
        // → `2**(k/2) <= sqrt(a) < 2**((k+1)/2) <= 2**(k/2 + 1)`
        //
        // Consequently, `2**(log2(a) / 2)` is a good first approximation of `sqrt(a)` with at least 1 correct bit.
        uint256 result = 1 << (log2(a) >> 1);

        // At this point `result` is an estimation with one bit of precision. We know the true value is a uint128,
        // since it is the square root of a uint256. Newton's method converges quadratically (precision doubles at
        // every iteration). We thus need at most 7 iteration to turn our partial result with one bit of precision
        // into the expected uint128 result.
        unchecked {
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            return min(result, a / result);
        }
    }

    /**
     * @notice Calculates sqrt(a), following the selected rounding direction.
     */
    function sqrt(uint256 a, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = sqrt(a);
            return result + (rounding == Rounding.Up && result * result < a ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 2, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 128;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 64;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 32;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 16;
            }
            if (value >> 8 > 0) {
                value >>= 8;
                result += 8;
            }
            if (value >> 4 > 0) {
                value >>= 4;
                result += 4;
            }
            if (value >> 2 > 0) {
                value >>= 2;
                result += 2;
            }
            if (value >> 1 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 2, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log2(value);
            return result + (rounding == Rounding.Up && 1 << result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 10, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >= 10**64) {
                value /= 10**64;
                result += 64;
            }
            if (value >= 10**32) {
                value /= 10**32;
                result += 32;
            }
            if (value >= 10**16) {
                value /= 10**16;
                result += 16;
            }
            if (value >= 10**8) {
                value /= 10**8;
                result += 8;
            }
            if (value >= 10**4) {
                value /= 10**4;
                result += 4;
            }
            if (value >= 10**2) {
                value /= 10**2;
                result += 2;
            }
            if (value >= 10**1) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log10(value);
            return result + (rounding == Rounding.Up && 10**result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 256, rounded down, of a positive value.
     * Returns 0 if given 0.
     *
     * Adding one to the result gives the number of pairs of hex symbols needed to represent `value` as a hex string.
     */
    function log256(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 16;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 8;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 4;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 2;
            }
            if (value >> 8 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log256(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log256(value);
            return result + (rounding == Rounding.Up && 1 << (result * 8) < value ? 1 : 0);
        }
    }
}

// File: contracts/helpers/Mocks/IWETH.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity >=0.5.0;

interface IWETH {
    function deposit() external payable;

    function transfer(address to, uint256 value) external returns (bool);

    function withdraw(uint256) external;
}
// File: contracts/staking/IMYCStakingFactory.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.7;

interface IMYCStakingFactory {
    function treasury() external view returns (address);

    function signer() external view returns (address);

    function mycStakingManager() external view returns (address);

    function owner() external view returns (address);

    function WETH() external view returns (address);
}

// File: contracts/staking/IMYCStakingManager.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.7;

interface IMYCStakingManager {
    function addStakingPool(address poolAddress, bytes32 signature) external;

    function treasury() external view returns (address);

    function signer() external view returns (address);

    function owner() external view returns (address);
}

// File: contracts/staking/IMYCStakingPool.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.7;

interface IMYCStakingPool {
    function factory() external returns (address);
}

// File: contracts/staking/LockedStaking/LockedStaking.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.7;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "../IMYCStakingFactory.sol";
import "../IMYCStakingPool.sol";
import "../../helpers/Mocks/IWETH.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";
import "@openzeppelin/contracts/utils/cryptography/draft-EIP712.sol";
import "@openzeppelin/contracts/utils/cryptography/ECDSA.sol";

struct StakingPool {
    address tokenAddress; // staking token address
    address owner; // owner of pool
    uint256 dateStart; // start date for all pools
    uint256 dateEnd; // end date for all pools
    bool rescued; // if unused rewards are withdrawn by owner
    uint256 mycFeesWithdrawn; // withdrawn myc fees
}

struct StakingPlan {
    uint256 duration; // for how long user cannot unstake (seconds)
    uint256 maxTokensBeStaked; // maximum amount that can be staked amoung all stakers
    uint256 availableTokensBeStaked; // available tokens amount can be staked by user
    uint256 rewardsPool; // reward pool
    uint256 rewardsWithdrawn; // how many rewards withdrawn by stakers
    uint256 mycFeesPool; //myc fees pools
    uint256 maxStakingAmount; //max staking amount
    bool isSignatureRequired; // if signature is required
}

struct UserStake {
    uint256 stakeDate;
    uint256 amount;
}

/// @title Locked Staking by MyCointainer
/// @notice Stake ERC20 tokens for rewards
contract LockedStaking is IMYCStakingPool, ReentrancyGuard, EIP712 {
    using SafeERC20 for IERC20;
    /**
     * @dev Emitted when `staker` stakes tokens for `stakingPlanId`
     */
    event Staked(
        address indexed staker,
        uint256 indexed stakingPlanId,
        uint256 amount,
        uint256 unstakeDate
    );

    event StakedWithSignature(
        address indexed staker,
        uint256 indexed stakingPlanId,
        uint256 amount,
        uint256 unstakeDate,
        uint256 nonce
    );

    /**
     * @dev Emitted when `staker` unstakes tokens for `stakingPlanId`
     */
    event Unstaked(
        address indexed staker,
        uint256 indexed stakingPlanId,
        uint256 amount,
        uint256 reward
    );

    /**
     * @dev Emitted when `staker` unstakes tokens wtih penalty(before staking period ends)
     */
    event UnstakedWithPenalty(
        address indexed staker,
        uint256 indexed stakingPlanId,
        uint256 staked,
        uint256 withdrawn
    );

    error OnlyFactory();
    error StakeAlreadyExists();
    error NoSlotsAvailable();
    error StakeNotFound();
    error EndOfStakeNotReached();
    error StakingPeriodNotEnded();
    error NothingToWithdraw();
    error DateInFuture();
    error DateInPast();
    error NoTokensAvailableToStake();
    error AmountCantBeZero();
    error AlreadyRescued();
    error MaxStakingAmountExceed();

    StakingPool internal _stakePool;
    mapping(address => mapping(uint256 => UserStake)) internal _userStake;
    StakingPlan[] internal _plans;
    IMYCStakingFactory internal _factory;
    mapping(uint256 => bool) public isNonceUsed;

    /**
     * @dev Plans length
     */
    function getPlansLength() external view returns (uint256) {
        return _plans.length;
    }

    /**
     * @dev Returns factory address
     */
    function factory() external view returns (address) {
        return address(_factory);
    }

    /**
     * @dev Returns plans
     */
    function getPlans() external view returns (StakingPlan[] memory) {
        StakingPlan[] memory plans = new StakingPlan[](_plans.length);
        plans = _plans;
        return plans;
    }

    /**
     * @dev Returns `planId` plan
     */
    function getPlan(
        uint256 planId
    ) external view returns (StakingPlan memory) {
        return _plans[planId];
    }

    /**
     * @dev Returns {StakingPool} info
     */
    function stakingPool() external view returns (StakingPool memory) {
        return _stakePool;
    }

    /**
     * @dev Returns stake info for each plan for staker
     */
    function stakesInfoOf(
        address staker
    ) external view returns (UserStake[] memory) {
        uint256 len = _plans.length;
        UserStake[] memory stakes = new UserStake[](len);
        for (uint256 i = 0; i < len; i++) {
            stakes[i] = _userStake[staker][i];
        }
        return stakes;
    }

    /**
     * @dev Returns stake info for `staker` and `planId`
     */
    function stakeInfoOf(
        address staker,
        uint256 planId
    ) external view returns (UserStake memory) {
        return _userStake[staker][planId];
    }

    constructor(
        address tokenAddress, // staking token address
        address owner, // owner of pool
        uint256 dateStart, // start date for all pools
        uint256 dateEnd, // end date for all pools
        StakingPlan[] memory plans //staking plans
    ) EIP712("MyCointainer", "1") {
        //saving pool info
        StakingPool memory newConfig = StakingPool({
            tokenAddress: tokenAddress,
            owner: owner,
            dateStart: dateStart,
            dateEnd: dateEnd,
            rescued: false,
            mycFeesWithdrawn: 0
        });
        _stakePool = newConfig;

        //saving _plans
        uint256 len = plans.length;
        for (uint256 i = 0; i < len; i++) {
            _plans.push(plans[i]);
        }

        _factory = IMYCStakingFactory(msg.sender);
    }

    function _checkDateInFutureOrZero(uint256 date) internal view {
        if (date == 0) {
            return;
        }
        if (date < block.timestamp) revert DateInPast();
    }

    function _withdrawTokensFromContract(
        address[] memory to,
        uint256[] memory amount
    ) internal {
        require(to.length == amount.length, "Length mismatch");
        IERC20 stakeToken = IERC20(_stakePool.tokenAddress);
        address wethAddress = IMYCStakingFactory(_factory).WETH();
        uint256 sum;
        for (uint256 i = 0; i < to.length; i++) {
            sum += amount[i];
        }
        if (wethAddress == address(stakeToken)) {
            IWETH(wethAddress).withdraw(sum);
        }
        for (uint256 i = 0; i < to.length; i++) {
            if (wethAddress == address(stakeToken)) {
                payable(to[i]).transfer(amount[i]);
            } else {
                stakeToken.safeTransfer(to[i], amount[i]);
            }
        }
    }

    function _depositTokensToContract(uint256 amount) internal {
        IERC20 stakeToken = IERC20(_stakePool.tokenAddress);
        address wethAddress = IMYCStakingFactory(_factory).WETH();
        if (wethAddress == address(stakeToken)) {
            require(amount == msg.value, "Native currency mismatch");
            IWETH(wethAddress).deposit{value: amount}();
        } else {
            stakeToken.safeTransferFrom(msg.sender, address(this), amount);
        }
    }

    function _checkDateInPastOrZero(uint256 date) internal view {
        if (date == 0) {
            return;
        }
        if (date > block.timestamp) revert DateInFuture();
    }

    function _checkPlanAvailability(
        StakingPlan memory plan,
        uint256 amountToStake
    ) internal pure returns (uint256) {
        if (plan.availableTokensBeStaked == 0) {
            revert NoTokensAvailableToStake();
        } else if (plan.availableTokensBeStaked < amountToStake) {
            return plan.availableTokensBeStaked;
        } else {
            return amountToStake;
        }
    }

    function stakeWithSignature(address staker, uint256 amount, uint256 stakingPlanId, uint256 nonce, uint256 deadline, bytes memory signature) external payable {
        StakingPlan memory plan = _plans[stakingPlanId];
        require(deadline == 0 || deadline >= block.timestamp, "LockedStaking: Expired deadline");
        require(!isNonceUsed[nonce] && staker != address(0), "LockedStaking: nonce already used");
        bytes32 typedHash = _hashTypedDataV4(
            keccak256(
                abi.encode(
                    keccak256(
                        "SignedEntrance(address staker,uint256 stakingPlanId,uint256 nonce,uint256 deadline)"
                    ),
                    staker == address(0) ? address(0) : msg.sender, 
                    stakingPlanId,
                    nonce,
                    deadline
                )
            )
        );
        if(staker != address(0)){
            isNonceUsed[nonce] = true;
        }
        require (ECDSA.recover(typedHash, signature) == _factory.signer(), "LockedStaking: Invalid signature");
        _stake(amount, stakingPlanId);

        emit StakedWithSignature(msg.sender, stakingPlanId, amount, block.timestamp + plan.duration, nonce);
    }

    function _stake(uint256 amount, uint256 stakingPlanId) internal nonReentrant{
        if (amount == 0) {
            revert AmountCantBeZero();
        }
        UserStake memory uStake = _userStake[msg.sender][stakingPlanId];

        //only one active stake for each plan per staker
        if (uStake.stakeDate != 0) revert StakeAlreadyExists();

        StakingPool memory sp = _stakePool;

        //check start and end date
        _checkDateInPastOrZero(sp.dateStart);
        _checkDateInFutureOrZero(sp.dateEnd);

        //check availability to stake such amount, decreasing amount to max available
        StakingPlan memory plan = _plans[stakingPlanId];
        amount = _checkPlanAvailability(plan, amount);

        //check max staking amount
        if (amount > plan.maxStakingAmount) {
            revert MaxStakingAmountExceed();
        }

        //transfering tokens to smart contract - allowance needed
        _depositTokensToContract(amount);

        //store stake data for user
        _userStake[msg.sender][stakingPlanId] = UserStake({
            stakeDate: block.timestamp,
            amount: amount
        });

        //decreasing available amount to stake for plan
        _plans[stakingPlanId].availableTokensBeStaked -= amount;
    }

    /**
     * @dev Stakes tokens
     */
    function stake(uint256 amount, uint256 stakingPlanId) external payable{
        StakingPlan memory plan = _plans[stakingPlanId];
        require(!plan.isSignatureRequired, "LockedStaking: Signature required"); 
        _stake(amount, stakingPlanId);

        //emit event
        emit Staked(
            msg.sender,
            stakingPlanId,
            amount,
            block.timestamp + plan.duration
        );
    }

    function _calculateReward(
        StakingPlan memory plan,
        uint256 stakedOnPlan
    ) internal pure returns (uint256) {
        return (plan.rewardsPool * stakedOnPlan) / plan.maxTokensBeStaked;
    }

    /**
     * @dev Unstakes tokens on selected `poolIndex`
     */
    function unstake(uint256 stakingPlanId) external nonReentrant{
        _unstake(stakingPlanId);
    }


    function _unstake(uint256 stakingPlanId) internal {
        UserStake memory uStake = _userStake[msg.sender][stakingPlanId];

        // checking is stake exist
        if (uStake.stakeDate == 0) revert StakeNotFound();

        // check end staking date
        StakingPlan memory plan = _plans[stakingPlanId];
        if (uStake.stakeDate + plan.duration > block.timestamp)
            revert EndOfStakeNotReached();

        // reset stake values for user
        _userStake[msg.sender][stakingPlanId] = UserStake({
            stakeDate: 0,
            amount: 0
        });

        // update withrawn rewards
        uint256 rewardToWithdraw = _calculateReward(plan, uStake.amount);
        _plans[stakingPlanId].rewardsWithdrawn += rewardToWithdraw;

        address[] memory addresses = new address[](1);
        uint256[] memory amounts = new uint256[](1);
        addresses[0] = msg.sender;
        amounts[0] = uStake.amount + rewardToWithdraw;

        // transfer tokens
        _withdrawTokensFromContract(addresses, amounts);

        // emit event
        emit Unstaked(
            msg.sender,
            stakingPlanId,
            uStake.amount,
            rewardToWithdraw
        );
    }

    /**
     * @dev Unstakes tokens on selected `poolIndex` before end of staking period
     * Takes 10% fee
     */
    function unstakeWithPenalty(uint256 stakingPlanId) external nonReentrant{
        UserStake memory uStake = _userStake[msg.sender][stakingPlanId];

        // checking is stake exist
        if (uStake.stakeDate == 0) revert StakeNotFound();

        // if after locked period - do normal unstake
        StakingPlan memory plan = _plans[stakingPlanId];
        if (uStake.stakeDate + plan.duration < block.timestamp) {
            _unstake(stakingPlanId);
            return ();
        }

        StakingPool memory sc = _stakePool;
        uint256 feeAmount = (uStake.amount * 5) / 100;
        uint256 toWithdraw = uStake.amount - 2 * feeAmount;

        // reset stake values for user
        _userStake[msg.sender][stakingPlanId] = UserStake({
            stakeDate: 0,
            amount: 0
        });

        // calculate and transfer tokens
        uint256 rescuedRewards = 0;
        if (sc.dateEnd > block.timestamp || sc.dateEnd == 0) {
            _plans[stakingPlanId].availableTokensBeStaked += uStake.amount;
        } else {
            rescuedRewards = _calculateReward(plan, uStake.amount);
            _plans[stakingPlanId].rewardsWithdrawn += rescuedRewards;
        }
        address[] memory addresses = new address[](3);
        uint256[] memory amounts = new uint256[](3);
        addresses[0] = sc.owner;
        addresses[1] = _factory.treasury();
        addresses[2] = msg.sender;
        amounts[0] = feeAmount + rescuedRewards;
        amounts[1] = feeAmount;
        amounts[2] = toWithdraw;
        _withdrawTokensFromContract(addresses, amounts);
        emit UnstakedWithPenalty(
            msg.sender,
            stakingPlanId,
            uStake.amount,
            toWithdraw
        );
    }

    /**
     * @dev Sends unused reward tokens back to owner
     *
     * Note: Can be used only after stake period end
     */
    function claimUnusedRewards() external nonReentrant{
        StakingPool memory sc = _stakePool;
        if (sc.dateEnd >= block.timestamp || sc.dateEnd == 0) {
            revert StakingPeriodNotEnded();
        }

        if (sc.rescued) {
            revert AlreadyRescued();
        }

        uint256 sumToRescue;
        for (uint256 i = 0; i < _plans.length; i++) {
            StakingPlan memory plan = _plans[i];
            sumToRescue +=
                (plan.availableTokensBeStaked * (plan.rewardsPool)) /
                plan.maxTokensBeStaked;
        }
        if (sumToRescue == 0) revert NothingToWithdraw();
        _stakePool.rescued = true;

        address[] memory addresses = new address[](1);
        uint256[] memory amounts = new uint256[](1);
        addresses[0] = sc.owner;
        amounts[0] = sumToRescue;

        // transfer tokens
        _withdrawTokensFromContract(addresses, amounts);
    }

    receive() external payable {
        assert(msg.sender == IMYCStakingFactory(_factory).WETH()); // only accept ETH via fallback from the WETH contract
    }
}

// File: contracts/staking/LockedStaking/LockedStakingFactory.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.7;

import "./LockedStaking.sol";
import "../IMYCStakingManager.sol";
import "../IMYCStakingFactory.sol";
import "../../helpers/Mocks/IWETH.sol";

/// @title Locked Staking Factory
/// @notice Creates new LockedStaking Contracts
contract LockedStakingFactory is EIP712, IMYCStakingFactory {
    /**
     * @dev Emitted when withdawing MYC `reward` fees for `poolAddress`
     */
    event WithdrawnMYCFees(address indexed poolAddress, uint256 reward);

    bytes32 public constant STAKING_PLAN_TYPEHASH =
        keccak256(
            "StakingPlan(uint256 duration,uint256 maxTokensBeStaked,uint256 availableTokensBeStaked,uint256 rewardsPool,uint256 rewardsWithdrawn,uint256 mycFeesPool,uint256 maxStakingAmount,bool isSignatureRequired)"
        );

    bytes32 public constant LOCKED_STAKING_POOL_TYPEHASH =
        keccak256(
            "LockedStakingPool(address poolOwner,address tokenAddress,address referralAddress,uint256 feeToReferral,uint256 dateStart,uint256 dateEnd,uint256 deadline,StakingPlan[] stakingPlans)StakingPlan(uint256 duration,uint256 maxTokensBeStaked,uint256 availableTokensBeStaked,uint256 rewardsPool,uint256 rewardsWithdrawn,uint256 mycFeesPool,uint256 maxStakingAmount,bool isSignatureRequired)"
        );

    struct LockedStakingPool {
        address poolOwner;
        address tokenAddress;
        address referralAddress;
        uint256 feeToReferral;
        uint256 dateStart;
        uint256 dateEnd;
        uint256 deadline;
        StakingPlan[] stakingPlans;
    }

    error WrongExecutor();

    IMYCStakingManager internal _mycStakingManager;
    IWETH internal _WETH;

    constructor(
        IMYCStakingManager mycStakingManager_,
        IWETH weth_
    ) EIP712("MyCointainer", "1") {
        _mycStakingManager = mycStakingManager_;
        _WETH = weth_;
    }

    /**
     * @dev Returns WETH address
     *
     */
    function WETH() external view returns (address) {
        return address(_WETH);
    }

    /**
     * @dev Returns MyCointainer Staking Manager Contract Address
     *
     */
    function mycStakingManager() external view returns (address) {
        return address(_mycStakingManager);
    }

    /**
     * @dev Returns signer address
     */
    function signer() external view returns (address) {
        return _mycStakingManager.signer();
    }

    /**
     * @dev Returns signer address
     */
    function treasury() external view returns (address) {
        return _mycStakingManager.treasury();
    }

    /**
     * @dev Returns main owner address
     */
    function owner() external view returns (address) {
        return _mycStakingManager.owner();
    }

    /**
     * @dev Creates {LockedStaking} new smart contract
     *
     */
    function createPool(
        LockedStakingPool calldata stakingPool,
        bytes calldata signature
    ) external payable {
        //check pool owner
        require(
            stakingPool.poolOwner == msg.sender ||
                stakingPool.poolOwner == address(0),
            "LSF: Wrong executor"
        );

        // checking dates
        require(
            stakingPool.dateEnd >= stakingPool.dateStart,
            "LSF: Dates sort"
        );

        require(
            block.timestamp <= stakingPool.deadline,
            "LSF: Transaction overdue"
        );

        _verify(stakingPool, signature);

        LockedStaking createdPool = new LockedStaking{salt: bytes32(signature)}(
            stakingPool.tokenAddress,
            msg.sender,
            stakingPool.dateStart,
            stakingPool.dateEnd,
            stakingPool.stakingPlans
        );

        _transferTokens(stakingPool, address(createdPool));

        _mycStakingManager.addStakingPool(
            address(createdPool),
            bytes32(signature)
        );
    }

    function _transferTokens(
        LockedStakingPool calldata stakingPool,
        address poolAddress
    ) internal {
        uint256 rewardPoolSum = 0;
        uint256 mycFeeSum = 0;
        for (uint256 i = 0; i < stakingPool.stakingPlans.length; i++) {
            mycFeeSum += stakingPool.stakingPlans[i].mycFeesPool;
            rewardPoolSum += stakingPool.stakingPlans[i].rewardsPool;
        }

        require(stakingPool.feeToReferral <= 10000, "Fee too high");
        if (stakingPool.feeToReferral > 0) {
            require(
                stakingPool.referralAddress != address(0),
                "Referral address is zero"
            );
        }
        uint256 feeToReferralWei = (stakingPool.feeToReferral * mycFeeSum) /
            10000;

        if (address(_WETH) == stakingPool.tokenAddress) {
            require(
                rewardPoolSum + mycFeeSum == msg.value,
                "Native currency amount mismatch"
            );
            _WETH.deposit{value: msg.value}();
            _WETH.transfer(poolAddress, rewardPoolSum);
            if (mycFeeSum > 0) {
                _WETH.transfer(
                    _mycStakingManager.treasury(),
                    mycFeeSum - feeToReferralWei
                );
            }
            if (feeToReferralWei > 0) {
                _WETH.transfer(stakingPool.referralAddress, feeToReferralWei);
            }
        } else {
            IERC20(stakingPool.tokenAddress).transferFrom(
                msg.sender,
                poolAddress,
                rewardPoolSum
            );

            if (feeToReferralWei > 0) {
                IERC20(stakingPool.tokenAddress).transferFrom(
                    msg.sender,
                    stakingPool.referralAddress,
                    feeToReferralWei
                );
            }

            if (mycFeeSum > 0) {
                IERC20(stakingPool.tokenAddress).transferFrom(
                    msg.sender,
                    _mycStakingManager.treasury(),
                    mycFeeSum - feeToReferralWei
                );
            }
        }
    }

    function _verify(
        LockedStakingPool calldata pool,
        bytes calldata signature
    ) internal view {
        require(
            _recoverAddressOfPool(pool, signature) ==
                _mycStakingManager.signer(),
            "LSF: Invalid signature"
        );
    }

    function _recoverAddressOfPool(
        LockedStakingPool calldata pool,
        bytes calldata signature
    ) internal view returns (address) {
        bytes32 typedHash = _hashTypedDataV4(keccak256(_encodePool(pool)));
        return ECDSA.recover(typedHash, signature);
    }

    function _recoverAddressOfPlan(
        StakingPlan calldata plan,
        bytes calldata signature
    ) internal view returns (address) {
        bytes32 typedHash = _hashTypedDataV4(keccak256(_encodePlan(plan)));
        return ECDSA.recover(typedHash, signature);
    }

    function _encodePlan(
        StakingPlan calldata plan
    ) internal pure returns (bytes memory) {
        return
            abi.encode(
                STAKING_PLAN_TYPEHASH,
                plan.duration,
                plan.maxTokensBeStaked,
                plan.availableTokensBeStaked,
                plan.rewardsPool,
                plan.rewardsWithdrawn,
                plan.mycFeesPool,
                plan.maxStakingAmount,
                plan.isSignatureRequired
            );
    }

    function _encodePool(
        LockedStakingPool calldata pool
    ) internal pure returns (bytes memory) {
        bytes32[] memory encodedPlans = new bytes32[](pool.stakingPlans.length);
        for (uint256 i = 0; i < pool.stakingPlans.length; i++) {
            encodedPlans[i] = keccak256(_encodePlan(pool.stakingPlans[i]));
        }
        return
            abi.encode(
                LOCKED_STAKING_POOL_TYPEHASH,
                pool.poolOwner,
                pool.tokenAddress,
                pool.referralAddress,
                pool.feeToReferral,
                pool.dateStart,
                pool.dateEnd,
                pool.deadline,
                keccak256(abi.encodePacked(encodedPlans))
            );
    }
}